# BÀI THỰC HÀNH 4 Mảng Động, Kiểu Date, Mảng Control, TextBox, ListBox, ComboBox

# 1. Hãy <u>ĐỌC CẨN THẬN TẤT CẢ NỘI DUNG</u> trong bài thực hành trước khi làm bài. 2. Sinh viên <u>TỰ GÕ CODE, KHÔNG COPY</u> từ bài có sẵn, để nắm được việc lập trình.

### 1 Chuẩn đầu ra

Sau bài này, người học có thể:

- Sử dụng được mảng động, nêu được sự khác nhau giữa *ReDim* và *ReDim Preserve*.
- Sử dụng được kiểu dữ liệu ngày tháng (date).
- Tạo và sử dụng được mảng các điều khiển (mảng control).
- Sử dụng được các control: TextBox, ListBox và ComboBox.

#### 2 Chuẩn bị

Đọc lại về Mảng động, Kiểu dữ liệu ngày tháng (Date), Mảng các điều khiển (mảng control) và Các control: TextBox, ListBox và ComboBox.

*Tài liệu tham khảo gọi ý:* Chương 5, Chương 6, Giáo trình Visual Basic 6.0, Nguyễn Đăng Quang, Đại học Sư Phạm Kỹ Thuật TP. Hồ Chí Minh.

#### 3 Phương tiện

Máy tính có cài đặt Visual Studio 6.0.

#### 4 Kiến thức bổ trợ

• *Mảng động*: Mảng có số phần tử có thể thay đổi được.

Để sử dụng mảng động, ta thực hiện 2 bước:

1. Khai báo mảng (không khai báo chỉ số phần tử lớn nhất). Ví dụ:

Dim MangA() as Integer

Khai báo chỉ số phần tử lớn nhất: dùng ReDim (các phần tử đang có trong mảng sẽ bị xóa) hoặc ReDim Preserve (các phần tử đang có trong mảng được giữ nguyên). Ví dụ:

ReDim MangA(5) 'Khai báo mảng MangA có 6 phần tử (có chỉ số từ 0 → 5), các phần tử đã có trong MangA bị xóa.
ReDim Preserve MangA(7) 'Khai báo mảng MangA có 8 phần tử (có chỉ số từ 0 → 7), các phần tử đã có trong MangA không bị xóa.

- Mång các điều khiển (mång control):
  - Là một mảng các điều khiển cùng loại. Ví dụ: mảng các TextBox, mảng các OptionButton
  - Để tạo mảng control:
    - Tạo một control và chỉnh sửa hoàn chỉnh các thuộc tính của nó (name, kích cỡ, caption...).
    - Copy control này (Ctrl-C) và paste lên form (Ctrl-V). Khi VB thông báo hỏi tạo mảng control thì chọn Yes:



Khi copy xong, ta sẽ thấy tên các control được VB đánh thêm chỉ số:

Properties - txtA	(0)	×	Properties - txtA	(1)	×
txtA(0) TextBox	¢.	-	txtA(1) TextBo	x	-
Alphabet	egorized		Alphabet	tegorized	
MousePointer	0 - Default		MousePointer	0 - Default	
MultiLine	False		MultiLine	False	
OLEDragMode	0 - Manual		OLEDragMode	0 - Manual	

Và khi viết code, để gọi các control này, ta *phải chỉ định thêm chỉ số* của nó. Ví dụ:

txtA(1).Text = "48"

- Các thuộc tính của TextBox:
  - SelStart: xác định vị trí bắt đầu được *quét chọn* trong nội dung Text của TextBox. Vị trí trong Text của TextBox được *tính từ 0* (ký tự đầu tiên của chuỗi trong Text có vị trí 0).
  - SelLength: xác định độ dài chuỗi muốn quét chọn (tính từ vị trí SelStart).



• SetFocus: đặt con nháy vào TextBox (sẵn sàng cho người dùng nhập liệu).

*3 loại ComboBox:* Thuộc tính Style của ComboBox quy định loại ComboBox. Có 3 loại khác nhau:



- Dropdown Combo: Cho phép *chọn* giá trị có sẵn *hoặc gõ* giá trị khác.
- Simple Combo: *Hiện list* các lựa chọn. Lưu ý: *phải chỉnh lại chiều cao* (height) của ComboBox mới thấy list lựa chọn.
- Dropdown List: Chỉ được chọn trong các giá trị có sẵn (không thể gõ giá trị khác).
- Các thuộc tính và hàm của ListBox:
  - Mỗi phần tử trong ListBox được gán một chỉ số (index), *tính từ 0*.
  - Mỗi phần tử trong ListBox sẽ có 2 giá trị: Chuỗi (lưu trong List) và Số (lưu trong *ItemData*). Giá trị Chuỗi sẽ được in ra trong ListBox, còn giá trị Số thì không.
    - List(*i*): lấy hoặc gán giá trị chuỗi cho phần tử có chỉ số *i*.
    - ItemDate(*i*): lấy hoặc gán giá trị số cho phần tử có chỉ số *i*.
  - NewIndex: thuộc tính này trả về chỉ số của phần tử vừa được thêm vào ListBox.
  - MultiSelect: chỉ định ListBox có cho phép chọn đồng thời nhiều phần tử hay không (multiple selection). Thuộc tính này có 3 giá trị:
    - None: không cho phép chọn đồng thời nhiều phần tử.
    - Simple: cho phép chọn nhiều phần tử. Chọn hoặc bỏ phần tử bằng cách nhấp chuột.
    - Extended: cho phép chọn nhiều phần tử. Để chọn nhiều phần tử, giữ phím Shift hoặc Ctrl và nhấp chuột.
  - ListIndex: chỉ số của phần tử đang được chọn. Nếu ListIndex = -1 nghĩa là không có phần tử nào đang được chọn. Điều này *chỉ đúng* với ListBox không cho phép chọn đồng thời nhiều phần tử (MultiSelect = None).

*Lưu ý:* Đối với ListBox cho phép chọn đồng thời nhiều phần tử (MultiSelect = Simple hoặc Extended), ListIndex *không* cho biết chỉ số các phần tử đang được chọn. Lúc này, ListIndex chỉ cho biết chỉ số của phần tử đang được focus (phần tử này có thể được chọn hoặc không) → ListIndex có thể *không có giá trị -1 khi không có phần tử nào được chọn*. Vì vậy để kiểm tra xem phần tử nào đang được chọn, ta dùng vòng lặp để kiểm tra thuộc tính Selected = True hay False của tất cả phần tử.

- Selected(*index*): lấy hoặc gán giá trị chỉ định trạng thái đang được chọn của phần tử có chỉ số *index* (đang được chọn: True, không được chọn: False).
   Ví dụ: lstA.Selected(3) = True → Phần tử có chỉ số 3 sẽ được chọn (tô xanh).
- TopIndex: lấy hoặc gán chỉ số của phần tử đang nằm ở trên cùng ListBox, thương dùng để cuộn ListBox.
- ListCount: tổng số phần tử có trong ListBox.
- Text: nội dung của phần tử đang được chọn. Thuộc tính này tương đương với kết của List(tenListBox.ListIndex).
- AddItem *string, index*: thêm một phần tử có nội dung *string* vào vị trí có chỉ số *index*. Nếu không chỉ định *index*, phần tử mới sẽ được thêm vào ở vị trí cuối.
- RemoveItem *index*: xóa phần tử có chỉ số *index*.
- Clear: xóa tất cả nội dung của ListBox.

# • Các hàm cần biết:

- Format (*number*, "*format*"): trả về giá trị (kiểu chuỗi) theo *format* quy định, ví dụ: lệnh Format (5.366, "#.##") → trả về chuỗi "5.37" (#.##: lấy 2 số thập phân). Nếu không có *format*, hàm Format() đơn giản trả về một chuỗi giống hàm Str().
- InputBox(): hàm hiện ra cửa sổ nhập liệu cho người dùng nhập dữ liệu và trả về dữ liệu được người dùng nhập vào.
- UBound(): trả về chỉ số lớn nhất (Upper Bound) của mảng. Lưu ý: chỉ số phần tử mảng bắt đầu bằng 0, do đó, chỉ số lớn nhất = số phần tử của mảng 1.

# Các hàm về ngày tháng:

- CDate(*string*): trả về dữ liệu kiểu Date của chuỗi *string* (Convert to Date).
- Date: trả về ngày giờ hiện tại (theo ngày giờ của máy).
- Day(date), Month(date), Year(date): trả về ngày, tháng, năm của ngày date. Ví dụ: Day(#22/1/2000#) → 22.
- Weekday(*date*): trả về thứ của ngày *date*, giá trị trả về từ 1-7, với 1: Chủ Nhật, 2: Thứ Hai... 7: Thứ Bảy.

# Các hàm về chuỗi:

- Len(*string*): trả về độ dài của chuỗi *string* (Length of string).
   Ví dụ: Len("CNTT SPKT") → 9
- Mid(*string, start\_position, length*): trả về chuỗi con của chuỗi *string* (Middle part of string). Chuỗi con được lấy từ vị trí bắt đầu *start\_position* và có độ dài *length*. Vị trí trong chuỗi được *tính từ 1* (ký tự đầu tiên của chuỗi có là *start\_position* 1). Ví dụ: Mid("Hom nay toi di hoc.", 5, 3) → "nay"
- LCase(*string*), UCase(*string*): trả về chuỗi *string* được in thường hoặc in hoa (lower**case**, UPPER**CASE**).

- Trim(*string*): trả về chuỗi *string* đã được cắt bỏ các khoảng trắng ở đầu và cuối chuỗi. Ví dụ: Trim(" Xin chao! ") → "Xin chao!"
- Space(number): trả về chuỗi có number khoảng trắng. Ví dụ: Space(6) → "

# 5 Thời lượng: 4 tiết

# 6 Nội dung thực hành

Bài thực hành có hướng dẫn chi tiết:

6.1	Sắp xếp các số lẻ giảm dần	5
6.2	Tính tuổi	9
6.3	Sắp xếp, tìm min-max, đếm loại số	11
6.4	Tìm kiếm trong chuỗi	15
6.5	Đặt món ăn dùng ListBox, ComboBox	21
Bài thụ	rc hành tự thực hiện:	
7.1	Bổ sung chức năng "Thay Thế" và "Thay Thế Tất Cả" cho bài 6.4	32
7.2	Không xem hướng dẫn chi tiết và tự thực hiện lại bài 6.5	32

# 6.1 SẮP XẾP CÁC SỐ LẢ GIẢM DẦN

# 6.1.1 Mô tả

Dữ liệu vào: mảng số nguyên.

Dữ liệu ra: mảng được sắp xếp: các số lẻ sắp giảm, các số chẵn giữ nguyên vị trí. Ví dụ: mảng nhập vào:  $\underline{13} 4 \overline{7} 8 12 \underline{93} - 668 \underline{37} \rightarrow \underline{379} 4 \overline{7} 8 12 \underline{33} - 668 \underline{1}$ .

Điều khiển:

- Nút *cmdNhap*: thực hiện nhập mảng và sắp xếp.
- Nút *cmdNhapThem*: thực hiện nhập thêm phần tử vào mảng và sắp xếp.

# 6.1.2 Thực hiện

0. Mở Visual Basic, chọn New Project/Standard EXE, và save vào thư mục Bai18\_SapXepCacSoLeGiamDan (tự tạo trước).

1. Tạo form như hình:

🔄. Sap cac so le giam dan	_ • •
Nhap so phan tu N: txtN	cmdNhap
Mang da nhap:	·····
:: IblMangDaN	hap :
Mang da sap xep cac so le giam d	an:
i IblMangDaSa	pXep :
Nhap so phan txtNthem tu muon them:	cmdNhapThem

Nhấp đúp vào nút *cmdNhap* và nhập code:
 Option Explicit

```
'Khai báo mảng động MangA()
Private MangA() As Integer
Private Sub HoanDoiViTri (a As Integer, b As Integer)
    Dim temp As Integer
    temp = a
    a = b
   b = temp
End Sub
Private Sub SapXepGiamDanCacSoLe(MangSoNguyen() As Integer)
    Dim ChiSoLonNhatCuaMang As Integer
    Dim i As Integer
    Dim j As Integer
    ChiSoLonNhatCuaMang = UBound (MangSoNguyen)
    For i = 0 To ChiSoLonNhatCuaMang - 1
        If MangSoNguyen(i) Mod 2 <> 0 Then
            For j = i + 1 To ChiSoLonNhatCuaMang
                If MangSoNguyen(j) Mod 2 <> 0 And MangSoNguyen(i)
< MangSoNguyen(j) Then
                    HoanDoiViTri MangSoNguyen(i), MangSoNguyen(j)
                End If
            Next
        End If
    Next
End Sub
```

```
Private Sub InMang(Mang() As Integer, lbl As Label)
       Dim i As Integer
       Dim ChuoiIn As String
       ChuoiIn = ""
       For i = 0 To UBound (Mang)
           ChuoiIn = ChuoiIn & " " & Mang(i)
       Next
       lbl.Caption = ChuoiIn
   End Sub
   Private Sub cmdNhap Click()
       Dim N As Integer
       Dim i As Integer
       N = Val(txtN.Text)
       'Chú ý: Sử dụng ReDim sẽ xóa tất cả các phần tử đã có
       trong mång.
       ReDim MangA(N - 1)
       For i = 0 To N - 1 Step 1
           MangA(i) = Val(InputBox("Nhap so thu " & i + 1))
       Next
       InMang MangA, lblMangDaNhap
       SapXepGiamDanCacSoLe MangA
       InMang MangA, lblMangDaSapXep
   End Sub
3. Nhấp đúp vào nút cmdNhapThem và nhập code:
   Private Sub cmdNhapThem Click()
       Dim Nthem As Integer
       Dim NtruocKhiThem As Integer
       Dim i As Integer
       Nthem = Val(txtNthem.Text)
       NtruocKhiThem = UBound(MangA)
       'Chú ý: Sử dụng ReDim Preserve để không xóa các phần tử
       đã có trong mảng.
       ReDim Preserve MangA (NtruocKhiThem + Nthem)
```

```
For i = NtruocKhiThem + 1 To Nthem + NtruocKhiThem Step 1
    MangA(i) = Val(InputBox("Nhap so thu " & i + 1))
Next
```

InMang MangA, lblMangDaNhap SapXepGiamDanCacSoLe MangA InMang MangA, lblMangDaSapXep End Sub

4. Lưu và nhấn nút Start hay F5 để chạy và xem kết quả:

🖪. Sap cac so le giam dan	
Nhap so phan tu N: 7	cmdNhap
Mang da nhap:	
IblMangDaNh	ap
Mang da sap xep cac so le giam dar	n:
IblMangDaSap	Хер
Nhap so phan tu muon them:	cmdNhapThem

Lần lượt nhập các phần tử vào:

Project1	×
Nhap so thu 4	OK Cancel
7	

Kết quả sau khi nhập xong:

🖪, Sap cac so le giar	m dan 🗖 🗖 🗙
Nhap so phan tu N:	7 [cmdNhap]
Mang da nhap:	
	13478129
Mang da sap xep ca	ac so le giam dan:
	97438121
Nhap so phan tu muon them:	cmdNhapThem

Nhập thêm 5 phần tử:

🔁 Sap cac so le giam dan
Nhap so phan tu N: 7 cmdNhap
Mang da nhap:
974381213-66837
Mang da sap xep cac so le giam dan:
37 9 4 7 8 12 3 3 -6 6 8 1
Nhap so phan 5 cmdNhapThem

# 6.2 **TÍNH TUỔI**

#### 6.2.1 Mô tả

Dữ liệu vào: ngày sinh.

Dữ liệu ra: số tuổi và thứ của ngày sinh. Ví dụ: Ngày sinh: 2/9/1945 → Tuổi: 67 tuổi (với năm hiện tại là 2012), ngày sinh là Chủ Nhật.

Điều khiển: nút cmdTinhTuoi: thực hiện tính tuổi và thứ của ngày sinh.

#### 6.2.2 Thực hiện

0. Mở Visual Basic, chọn New Project/Standard EXE, và save vào thư mục Bai19\_Date\_TinhTuoi (tự tạo trước).

1. Tạo form như hình:

🕄 Tinh tuoi		
Ngay sinh:	txtNgaySinh	cmdTinhTuoi
	Vidu: 1/1/2000 3-1-2012 Jan 7, 1990	
	IbIKQ	

2. Nhấp đúp vào nút *cmdTinhTuoi* và nhập code:

```
Option Explicit
```

```
Private Sub cmdTinhTuoi Click()
    Dim NgaySinh As Date
    Dim HomNay As Date
    Dim Tuoi As Byte
    Dim NgayTrongTuan As Byte
    'Kiểm tra dữ liệu nhập vào có phải là ngày tháng không.
    Nếu không thì dừng thủ tục.
    If IsDate(txtNgaySinh.Text) = False Then
        txtNqaySinh.Text = "Xin nhap mot ngay."
        txtNgaySinh.SelStart = 0
        txtNgaySinh.SelLength = Len(txtNgaySinh.Text)
        txtNgaySinh.SetFocus
                    'Dừng thủ tục.
        Exit Sub
    End If
    NgaySinh = CDate(txtNgaySinh.Text)
    HomNay = Date
    Tuoi = Year(HomNay) - Year(NgaySinh)
    NgayTrongTuan = Weekday(NgaySinh)
    lblKQ.Caption = "Ban " & Tuoi & " tuoi." & vbNewLine & "Ban
sinh vao " & Thu(NgayTrongTuan) & "."
End Sub
'Hàm đổi ngày trong tuần (số) thành thứ (chuỗi).
Ví dụ: 1 → Chủ Nhật, 2 → Thứ Hai, 3 → Thứ Ba...
Private Function Thu (NgayTrongTuan As Byte) As String
    Select Case NgayTrongTuan
    Case 1
        Thu = "Chu Nhat"
    Case 2
        Thu = "Thu Hai"
```

```
Case 3

Thu = "Thu Ba"

Case 4

Thu = "Thu Tu"

Case 5

Thu = "Thu Nam"

Case 6

Thu = "Thu Sau"

Case 7

Thu = "Thu Bay"

Case Else

Thu = "!!!Hay nhap NgayTrongTuan tu 1-7!!!"

End Select

End Function
```

3. Lưu và nhấn nút Start hay F5 để chạy và xem kết quả:

🔄, Tinh tuoi		
Ngay sinh:	1/1/2000 Vi du: 1/1/2000 3-1-2012 Jan 7, 1990	[ cmdTinhTuoi ]
	Ban 12 tuoi. Ban sinh ∨ao Thu B	ay.
🔄. Tinh tuoi		
Tinh tuoi Ngay sinh:	Sep 2, 1945 Vi du: 1/1/2000 3·1·2012 Jan 7, 1990	cmdTinhTuoi

# 6.3 SẮP XẾP, TÌM MIN-MAX, ĐẾM LOẠI SỐ

#### 6.3.1 Mô tả

Dữ liệu vào: mảng số nguyên.

Dữ liệu ra: mảng được sắp xếp tăng hoặc giảm, chỉ ra phần tử có giá trị lớn nhất và nhỏ nhất của mảng, liệt kê các số có trong mảng.

Điều khiển: nút *cmdNhap*: thực hiện nhập mảng, sắp xếp và tìm các dữ liệu ra.

### 6.3.2 Thực hiện

- 0. Mở Visual Basic, chọn New Project/Standard EXE, và save vào thư mục Bai20\_MangControl\_SapXep\_TimMinMax\_DemLoaiSo (tự tạo trước).
- 1. Tạo form như hình:

🔄 Mang control				- • •
Nhap mang so: txtA(0) txtA(1) txtA(2)	txtA(3) txtA(4) txtA(5)	) [txtA(6) [txtA(7)	txtA(8) [txtA(9)	IbILoaiSo
· · · Many sap glam dan. · · ·	lblSapGiamDa	an		
Mang sap tang dan:				
	lblSapTangDa	an		
	Min :	IblMax		

#### $Chú \ \acute{y}: txtA(0) - txtA(9)$ là mảng control.

2. Nhấp đúp vào nút *cmdNhap* và nhập code:

```
Option Explicit
Const SoPhanTu As Integer = 10
```

```
Private Function TimMax(A() As Double) As String
    Dim i As Integer
    Dim max As Double
    max = A(0)
    For i = 1 To SoPhanTu - 1
        If A(i) > max Then
            max = A(i)
        End If
    Next
    TimMax = Str(max)
End Function
Private Function TimMin(A() As Double) As String
    Dim i As Integer
    Dim min As Double
    min = A(0)
    For i = 1 To SoPhanTu - 1
        If A(i) < \min Then
            min = A(i)
        End If
    Next
```

```
TimMin = Str(min)
End Function
Private Function SapTangDan(A() As Double) As String
    Dim i As Integer, j As Integer
    Dim KQ As String
    For i = 0 To SoPhanTu - 2
        For j = i + 1 To SoPhanTu - 1
            If A(i) > A(j) Then
                HoanDoiViTri A(i), A(j)
            End If
        Next
    Next
    KQ = ""
    For i = 0 To SoPhanTu - 1
        KQ = KQ \& " " \& Str(A(i))
    Next
    SapTangDan = KQ
End Function
Private Function SapGiamDan(A() As Double) As String
    Dim i As Integer, j As Integer
    Dim KQ As String
    For i = 0 To SoPhanTu - 2
        For j = i + 1 To SoPhanTu - 1
            If A(i) < A(j) Then
                HoanDoiViTri A(i), A(j)
            End If
        Next
    Next
    KQ = ""
    For i = 0 To SoPhanTu - 1
        KQ = KQ \& " " \& Str(A(i))
    Next
    SapGiamDan = KQ
End Function
Private Sub HoanDoiViTri(a As Double, b As Double)
    Dim temp As Double
    temp = a
    a = b
    b = temp
End Sub
```

```
Private Function DemLoaiSo(A() As Double) As String
    Dim PhanTuDangDem As Double
    Dim SoLanXuatHien As Integer
    Dim i As Integer
    Dim KQ As String
    SapTangDan A
    PhanTuDangDem = A(0)
    SoLanXuatHien = 1
   KO = ""
    For i = 1 To SoPhanTu - 1
        If A(i) = PhanTuDangDem Then
            SoLanXuatHien = SoLanXuatHien + 1
        Else
            KQ = KQ & "Co " & Format(SoLanXuatHien) & " so " &
Format(PhanTuDangDem) & vbNewLine
            PhanTuDangDem = A(i)
            SoLanXuatHien = 1
        End If
    Next
    KQ = KQ & "Co " & Format (SoLanXuatHien) & " so " &
Format (PhanTuDangDem)
    DemLoaiSo = KQ
End Function
Private Sub cmdNhap_Click()
    Dim A(SoPhanTu - 1) As Double
    Dim i As Integer
    For i = 0 To SoPhanTu - 1
        A(i) = Val(txtA(i).Text)
    Next
    lblSapTangDan.Caption = SapTangDan(A)
    lblSapGiamDan.Caption = SapGiamDan(A)
    lblMin.Caption = "Phan tu nho nhat: " & TimMin(A)
    lblMax.Caption = "Phan tu lon nhat: " & TimMax(A)
    lblLoaiSo.Caption = DemLoaiSo(A)
End Sub
```

3. Lưu và nhấn nút Start hay F5 để chạy và xem kết quả:

Nhap mang so:       Co 1 so -6         2.14       3       4       6       -6       7       10.48       7       5.805       7       Co 1 so 2.14         Co 1 so 3       Co 1 so 3       Co 1 so 4       Co 1 so 5.805       Co 1 so 5.805       Co 1 so 5.805         Mang sap giam dan:       -6       2.14 3 4 5.805 6 7 7 7 10.48       Co 1 so 6       Co 3 so 7       Co 1 so 10.48         Mang sap tang dan:       10.48 7 7 7 6 5.805 4 3 2.14-6       Co 1 so 10.48       Co 1 so 10.48       Co 1 so 10.48	🔄. Mang control	
Phan tu nho nhat: -6 Phan tu Ion nhat: 10.48	Nhap mang so:         2.14       3       4       6       -6       7       10.48       7       5.805       7         cmdNhap         Mang sap giam dan:         -6       2.14       3       4       5.805       6       7       7       10.48         Mang sap giam dan:       -6       2.14       3       4       5.805       6       7       7       10.48         Mang sap tang dan:       10.48       7       7       6       5.805       4       3       2.14       -6         Phan tu nho nhat: -6       Phan tu lon nhat: 10.48       Phan tu lon nhat: 10.48       Phan tu lon nhat: 10.48	Co 1 so -6 Co 1 so 2.14 Co 1 so 3 Co 1 so 4 Co 1 so 5.805 Co 1 so 6 Co 3 so 7 Co 1 so 10.48

# 6.4 TÌM KIẾM TRONG CHUỗI

# 6.4.1 Mô tả

Dữ liệu vào: chuỗi văn bản.

Dữ liệu ra: một chuỗi cần tìm được định vị trong văn bản.

Điều khiển: nút *cmdTim*: thực hiện tìm kiếm:

- (1) Khi người dùng nhấn nút *cmdTim* lần đầu, chương trình sẽ tìm kiếm vị trí xuất hiện đầu tiên của chuỗi cần tìm.
- (2) Khi người dùng nhấn nút *cmdTim* từ lần thứ 2, chương trình sẽ thực hiện tìm kiếm vị trí xuất hiện tiếp theo của chuỗi.
- (3) Khi người dùng tiếp tục nhấn *cmdTim* nhiều lần cho đến khi đã tìm hết văn bản, chương trình sẽ thông báo *tổng số lần xuất hiện của chuỗi cần tìm* trong văn bản.
- (4) Khi người dùng nhấn nút *cmdTim* mà nội dung văn bản hoặc chuỗi cần tìm bị bỏ trống, chương trình sẽ nhắc người dùng nhập dữ liệu.
- (5) Khi người dùng thay đổi nội dung văn bản hoặc chuỗi cần tìm, việc tìm kiếm sẽ được bắt đầu lại từ đầu.

# 6.4.2 Thực hiện

0. Mở Visual Basic, chọn New Project/Standard EXE, và save vào thư mục Bai21\_TextBox\_TimKiemTrongChuoi (tự tạo trước).

1. Tạo form như hình:

3.	
txtChuoi MultiLine: True ScrollBars: Vertical	txtTim cmdTim IbIThongBao

Chú ý: TextBox txtChuoi có thuộc tính MultiLine: True và ScrollBars: Vertical.

2. Nhấp đúp vào nút *cmdTim* và nhập code:

```
Option Explicit
```

```
'Các biến được khai báo ở vị trí toàn cục để có thể dùng trong
các hàm/thủ tục khác nhau.
Private SoLanTimThay As Integer
Private ViTriDangTim As Integer
Private Sub cmdTim Click()
    Dim Chuoi As String
    Dim ChuoiTim As String
    Dim DoDaiChuoiTim As Integer
    Chuoi = LCase(txtChuoi.Text)
    ChuoiTim = LCase(txtTim.Text)
    DoDaiChuoiTim = Len(ChuoiTim)
    'Nhắc người dùng nhập liệu nếu chưa nhập (chức năng (4)).
    If (Len(Chuoi) = 0) Then
        lblThongBao.Caption = "Xin nhap van ban."
        lblThongBao.BackColor = vbRed
        txtChuoi.SetFocus
        Exit Sub
    End If
    If (DoDaiChuoiTim = 0) Then
        lblThongBao.Caption = "Xin nhap chuoi can tim."
        lblThongBao.BackColor = vbRed
        txtTim.SetFocus
        Exit Sub
    End If
```

```
'Thưc hiện tìm kiếm (chức năng (1) (2) và (3))
    Do While ViTriDangTim <= Len(Chuoi)</pre>
        If Mid(Chuoi, ViTriDangTim, DoDaiChuoiTim) = ChuoiTim
Then
            'Nếu tìm thấy thì quét chọn chuỗi cho người dùng thấy
            txtChuoi.SelStart = ViTriDangTim - 1
            txtChuoi.SelLength = DoDaiChuoiTim
            txtChuoi.SetFocus
            'Đặt lại ViTriDangTim cho lần tìm kiếm sau.
            ViTriDangTim = ViTriDangTim + DoDaiChuoiTim
            Exit Do
        Else
            'Nếu không tìm thấy thì tìm qua vị trí kế tiếp.
            ViTriDangTim = ViTriDangTim + 1
        End If
    Loop
    If (SoLanTimThay = 0) And ViTriDangTim > Len(Chuoi) Then
        'Thông báo khi không tìm thấy.
        lblThongBao.Caption = "Khong tim thay <" & ChuoiTim & ">
trong van ban."
        lblThongBao.BackColor = vbRed 'RGB(250, 0, 0)
    ElseIf ViTriDangTim > Len(Chuoi) Then
        'Thông báo khi đã tìm hết văn bản (chức năng (3)).
        lblThongBao.Caption = "Da tim toan bo van ban. Co tat ca
" & SoLanTimThay & " " & "<" & ChuoiTim & ">."
        lblThongBao.BackColor = vbWhite
                                          'RGB(255, 255,255)
        DatLaiCacGiaTri De BatDauTimKiemTuDau
    Else
        'Thông báo khi tìm thấy chuỗi (chức năng (1) (2)).
        lblThongBao.Caption = "Tim thay <" & ChuoiTim & "> thu "
& SoLanTimThay + 1 & "."
        lblThongBao.BackColor = vbGreen
                                         'RGB(0, 255, 0)
        SoLanTimThay = SoLanTimThay + 1
    End If
End Sub
'Đặt lại các giá trị tìm kiếm, để bắt đầu tìm từ đầu văn bản.
Private Sub DatLaiCacGiaTri De BatDauTimKiemTuDau()
    SoLanTimThay = 0
   ViTriDangTim = 1
End Sub
```

3. Nhấp đúp vào TextBox *txtChuoi* và nhập code:

```
'Tìm kiếm lại từ đầu khi người dùng thay đổi văn bản (chức năng
(5)).
Private Sub txtChuoi_Change()
    DatLaiCacGiaTri_De_BatDauTimKiemTuDau
End Sub
```

4. Nhấp đúp vào TextBox *txtTim* và nhập code:

```
'Tìm kiếm lại từ đầu khi người dùng thay đổi chuỗi cần tìm (chức
năng (5)).
Private Sub txtTim_Change()
DatLaiCacGiaTri_De_BatDauTimKiemTuDau
End Sub
```

5. Nhấp đúp vào *form* và nhập code:

```
'Khởi tạo các giá trị ban đầu cho biến SoLanTimThay và
ViTriDangTim khi form load.
```

```
Private Sub Form_Load()
    SoLanTimThay = 0
    ViTriDangTim = 1
End Sub
```

6. Có thể thêm lệnh sau trong sự kiện *form load* để nhập sẵn một đoạn văn bản cho việc tìm kiếm:

```
Private Sub Form Load()
```

```
SoLanTimThay = 0
ViTriDangTim = 1
```

txtChuoi.Text = "A prime number (or a prime) is a natural number greater than 1 that has no positive divisors other than 1 and itself. A natural number greater than 1 that is not a prime number is called a composite number. For example, 5 is prime because only 1 and 5 divide it, whereas 6 is composite because it has the divisors 2 and 3 in addition to 1 and 6. The fundamental theorem of arithmetic establishes the central role of primes in number theory: any integer greater than 1 can be expressed as a product of primes that is unique up to ordering. The uniqueness in this theorem requires excluding 1 as a prime because it is the multiplicative identity."

End Sub

7. Lưu và nhấn nút Start hay F5 để chạy và xem kết quả:

Nhấn nút *cmdTim* khi chưa nhập văn bản tìm kiếm (chức năng (4)) :

🕄. Tim kiem trong chuoi	
A	
	cmdTim
	Xin nhap van ban.
~	

Nhấn nút cmdTim khi chưa nhập chuỗi cần tìm (chức năng (4)):

🕄. Tim kiem trong chuoi	
A prime number (or a prime) is a natural number greater than 1 that has no positive divisors other than 1 and itself. A natural number greater than 1 that is not a prime number is called a composite number. For example, 5 is prime because only 1 and 5 divide it, whereas 6 is composite because it has the divisors 2 and 3 in addition to 1 and 6. The fundamental theorem of arithmetic establishes the central role of primes in number theory: any integer greater than 1 can be expressed as a product of primes that is unique up to	cmdTim Xin nhap chuoi can tim.

Nhập một từ có trong văn bản và nhấn *cmdTim* (chức năng (1)):

🔁. Tim kiem trong chuoi	
A prime number (or a prime) is a natural number greater than 1 that has no positive divisors other than 1 and itself. A natural number greater than 1 that is not a prime number is called a composite number. For example, 5 is prime because only 1 and 5 divide it, whereas 6 is composite because it has the divisors 2 and 3 in addition to 1 and 6. The fundamental theorem of arithmetic establishes the central role of primes in number theory: any integer greater than 1 can be expressed as a product of primes that is unique up to	Prime cmdTim Tim thay <prime> thu 1.</prime>

Tiếp tục nhấn cmdTim (chức năng (2)):

	_	
A prime number (or a prime) is a natural number greater than 1 that has no positive divisors other than 1 and itself. A natural number greater than 1 that is not a prime number is called a composite number. For example, 5 is prime because only 1 and 5 divide it, whereas 6 is composite because it has the divisors 2 and 3 in addition to 1 and 6. The fundamental theorem of arithmetic establishes the central role of primes in number theory: any integer greater than 1 can be	H	Prime cmdTim Tim thay <prime> thu 2.</prime>

Nhấn *cmdTim* nhiều lần cho đến khi đã tìm hết văn bản (chức năng (3)):

🔄. Tim kiem trong chuoi	
and itself. A natural number greater than 1 that is not a prime number is called a composite number. For example, 5 is prime because only 1 and 5 divide it, whereas 6 is composite because it has the divisors 2 and 3 in addition to 1 and 6. The fundamental theorem of arithmetic establishes the central role of primes in number theory: any integer greater than 1 can be expressed as a product of primes that is unique up to ordering. The uniqueness in this theorem requires excluding 1 as a prime because it is the multiplicative	Prime cmdTim Da tim toan bo van ban. Co tat ca 7 <prime>.</prime>

# Nhập một từ khác và nhấn cmdTim (chức năng (5)):

Tim kiem trong chuoi	
A prime number (or a prime) is a natural number greater than 1 that has no positive divisors other than 1 and itself. A natural number greater than 1 that is not a prime number is called a composite number. For example, 5 is prime because only 1 and 5 divide it, whereas 6 is composite because it has the divisors 2 and 3 in addition to 1 and 6. The fundamental theorem of arithmetic establishes the central role of primes in number theory: any integer greater than 1 can be expressed as a product of primes that is unique up to	number cmdTim Tim thay <number> thu 1.</number>

Nhập một từ không tồn tại trong văn bản:

13. Tim kiem trong chuoi	
A prime number (or a prime) is a natural number greater than 1 that has no positive divisors other than 1 and itself. A natural number greater than 1 that is not a prime number is called a composite number. For example, 5 is prime because only 1 and 5 divide it, whereas 6 is composite because it has the divisors 2 and 3 in addition to 1 and 6. The fundamental theorem of arithmetic establishes the central role of primes in number theory: any integer greater than 1 can be expressed as a product of primes that is unique up to	SPKT cmdTim Khong tim thay kspkt> trong van ban.

# 6.5 ĐẶT MÓN ĂN DÙNG LISTBOX, COMBOBOX

# 6.5.1 Mô tả

Dữ liệu vào: số lượng thực khách.

Dữ liệu ra: các món ăn được chọn, tổng số tiền phải trả.

🕄. Dat mon an	
So thuc khach: cboSoThucKhach 💌	
Thuc don: Gia (VND): Cac mon da chon:	
IstGia	CommandButton:
	cmdChonMon cmdChonTatCa
cmdHoanTat cmdChonLai line lblTongGiaTien	cmdBoChon
	cmdBoChonTatCa

Điều khiển:

- (1) Nhấp đúp vào một món ăn trong lstThucDon sẽ đưa món ăn vào lstMonDaChon và ngược lại.
- (2) Khi *nhấp chọn* một (hoặc nhiều) món ăn trong *lstThucDon*, giá của món ăn (hoặc các món ăn) tương ứng trong *lstGia* sẽ được chọn (tô sáng).
- (3) Khi *cuộn lstThucDon, lstGia* sẽ cuộn theo.
- (4) Nút cmdChonMon: di chuyển các món ăn được chọn từ lstThucDon sang lstMonDaChon. (5) Nút cmdBoChon: thực hiện chức năng ngược lại.
- (6) Nút *cmdChonTatCa*: *di chuyển tất cả* các món ăn sang *lstMonDaChon*. (7) Nút *cmdBoChonTatCa*: di chuyển tất cả các món ăn về *lstThucDon*.

- (8) Nút *cmdHoanTat*: *in thông báo* các món đã chọn, tổng số tiền và thoát chương trình.
- (9) Nút *cmdChonLai*: *trở lại trạng thái ban đầu* để thực hiện chọn lại các món ăn.
- Tất cả các thao tác trên đều làm cập nhật tổng số tiền phải thanh toán.

# 6.5.2 Thực hiện

- 0. Mở Visual Basic, chọn New Project/Standard EXE, và save vào thư mục Bai22\_ListBox\_ComboBox\_DatMonAn (tự tạo trước).
- 1. Tạo form như hình:

🕄 Dat mon an	
So thuc khach: cboSoThucKhach 💌	
Thuc don: Gia (VND): Cac mon da chon:	
IstThucDon IstGia IstGi	- CommandButton: cmdChonMon cmdChonTatCa
cmdHoanTat cmdChonLai IbITongGiaTien	cmdBoChon cmdBoChonTatCa

Chú ý:

- ComboBox *cboSoThucKhach* có thuộc tính *Style: Dropdown Combo* (vừa cho phép chọn trong các giá trị có sẵn, vừa cho gõ giá trị mới).
- ListBox *lstThucDon* và *lstGia* có thuộc tính *MultiSelect: Extended* (cho phép chọn đồng thời nhiều phần tử bằng cách giữ phím Shift hay Ctrl và nhấp chuột).
- ListBox *lstGia* có thuộc tính *Enabled: False* (chỉ để xem, không cho chọn).
- 2. Nhấp đúp vào *form* và nhập code:

```
Option Explicit
Private SoThucKhach As Byte
Private TongGiaTienChoMotNguoi As Long
'Các khởi tạo
Private Sub KhoiTao_cboSoThucKhach()
    Dim i As Byte
    For i = 1 To 10 Step 1
        cboSoThucKhach.AddItem Str(i)
    Next
        cboSoThucKhach.Text = "1"
End Sub
```

```
Private Sub KhoiTao_lstThucDon()
```

'Tên món ăn được lưu trong giá trị Chuỗi (List) của ListBox. 'Giá tiền của món ăn được lưu trong giá trị Số (ItemData) của ListBox.

```
lstThucDon.AddItem "Nam kho tieu"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 10000
    lstThucDon.AddItem "Nam chien gion"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 15000
    lstThucDon.AddItem "Dau hu khia nuoc dua"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 20000
    lstThucDon.AddItem "Dau hu chien xa"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 15000
    lstThucDon.AddItem "Khoai tay xao nam"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 20000
    lstThucDon.AddItem "Rau cu xao chua ngot"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 25000
    lstThucDon.AddItem "Rau muong xao"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 15000
    lstThucDon.AddItem "Canh chua"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 20000
    lstThucDon.AddItem "Canh nam"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 30000
    lstThucDon.AddItem "Canh rau cu"
    lstThucDon.ItemData(lstThucDon.NewIndex) = 25000
End Sub
```

#### Private Sub KhoiTao\_lstGia()

Dim i As Integer

```
For i = 0 To lstThucDon.ListCount - 1 Step 1
    lstGia.AddItem Str(lstThucDon.ItemData(i))
    Next
d Sub
```

```
End Sub
```

#### Private Sub Form\_Load()

```
SoThucKhach = 1
```

KhoiTao\_cboSoThucKhach KhoiTao\_lstThucDon KhoiTao\_lstGia End Sub

3. Nhấp đúp vào ComboBox *cboSoThucKhach* và nhập code:

```
Private Sub cboSoThucKhach_Change()
    If IsNumeric(cboSoThucKhach.Text) = False Then
        cboSoThucKhach.Text = ""
        SoThucKhach = 1
    Else
        SoThucKhach = Val(cboSoThucKhach.Text)
    End If
```

lblTongGiaTien.Caption = TongGiaTienChoMotNguoi \* SoThucKhach End Sub

```
Private Sub cboSoThucKhach_Click()
    SoThucKhach = Val(cboSoThucKhach.Text)
    lblTongGiaTien.Caption = TongGiaTienChoMotNguoi * SoThucKhach
End Sub
```

```
Private Sub cboSoThucKhach_LostFocus()
```

```
If cboSoThucKhach.Text = "" Then
    cboSoThucKhach.Text = "1"
End If
```

```
SoThucKhach = Val(cboSoThucKhach.Text)
lblTongGiaTien.Caption = TongGiaTienChoMotNguoi * SoThucKhach
End Sub
```

4. Nhấp đúp vào ListBox *lstThucDon* và nhập code:

```
Private Sub lstThucDon_Click()
```

```
Dim IndexMonAn As Integer
    'Chức năng (2): Tô chọn giá tiền các món ăn đang được chọn.
    IndexMonAn = 0
    Do While IndexMonAn <= (lstThucDon.ListCount - 1)</pre>
        If lstThucDon.Selected(IndexMonAn) = True Then
            lstGia.Selected(IndexMonAn) = True
        Else
            lstGia.Selected(IndexMonAn) = False
        End If
        IndexMonAn = IndexMonAn + 1
    Loop
    'Cuộn danh sách giá tiền cho tương ứng với danh sách món ăn.
    lstGia.TopIndex = lstThucDon.TopIndex
End Sub
Private Sub lstThucDon Scroll()
    'Chức năng (3): Cuộn danh sách giá tiền cho tương ứng với
danh sách món ăn.
    lstGia.TopIndex = lstThucDon.TopIndex
End Sub
'Chức năng (1): Nhấp đúp để chọn hoặc bỏ món.
Private Sub lstThucDon DblClick()
    lstMonDaChon.AddItem lstThucDon.List(lstThucDon.ListIndex)
    lstMonDaChon.ItemData(lstMonDaChon.NewIndex) =
lstThucDon.ItemData(lstThucDon.ListIndex)
    lstGia.RemoveItem lstThucDon.ListIndex
    lstThucDon.RemoveItem lstThucDon.ListIndex
```

```
TongGiaTienChoMotNguoi = TongGiaTienChoMotNguoi +
   lstMonDaChon.ItemData(lstMonDaChon.NewIndex)
       lblTongGiaTien.Caption = TongGiaTienChoMotNguoi * SoThucKhach
   End Sub
5. Nhấp đúp vào ListBox lstMonDaChon và nhập code:
   'Chức năng (1): Nhấp đúp để chọn hoặc bỏ món.
   Private Sub lstMonDaChon DblClick()
       lstThucDon.AddItem lstMonDaChon.List(lstMonDaChon.ListIndex)
       lstThucDon.ItemData(lstThucDon.NewIndex) =
   lstMonDaChon.ItemData(lstMonDaChon.ListIndex)
       lstGia.AddItem
   Str(lstMonDaChon.ItemData(lstMonDaChon.ListIndex))
       lstMonDaChon.RemoveItem lstMonDaChon.ListIndex
       TongGiaTienChoMotNguoi = TongGiaTienChoMotNguoi -
   lstThucDon.ItemData(lstThucDon.NewIndex)
       lblTongGiaTien.Caption = TongGiaTienChoMotNguoi * SoThucKhach
   End Sub
6. Nhấp đúp vào nút cmdChonMon và nhập code:
   Private Sub Tu ListBoxA Sang ListBoxB(lstA As ListBox, lstB As
   ListBox, ChonThem As Boolean)
       Dim Index 1stA As Integer
       Index lstA = 0
       Do While Index lstA <= (lstA.ListCount - 1)</pre>
           If lstA.Selected(Index lstA) = True Then
               lstB.AddItem lstA.List(Index lstA)
               lstB.ItemData(lstB.NewIndex) =
   lstA.ItemData(Index lstA)
               If ChonThem = True Then
                   TongGiaTienChoMotNguoi = TongGiaTienChoMotNguoi +
   lstA.ItemData(Index lstA)
                   lstGia.RemoveItem Index lstA
               Else
                   TongGiaTienChoMotNguoi = TongGiaTienChoMotNguoi -
   lstA.ItemData(Index lstA)
                    lstGia.AddItem Str(lstA.ItemData(Index lstA))
               End If
               lstA.RemoveItem Index lstA
               Index lstA = Index lstA - 1
           End If
           Index lstA = Index lstA + 1
       Loop
       lblTongGiaTien.Caption = TongGiaTienChoMotNguoi * SoThucKhach
```

```
'Chức năng (4): Chọn món.
   Private Sub cmdChonMon Click()
       Tu ListBoxA Sang ListBoxB lstThucDon, lstMonDaChon, True
   End Sub
7. Nhấp đúp vào nút cmdChonTatCa và nhập code:
   Private Sub ChuyenTatCa Tu ListBoxA Sang ListBoxB(lstA As
   ListBox, 1stB As ListBox, ChonThem As Boolean)
       Dim Index lstA As Integer
       For Index lstA = 0 To lstA.ListCount - 1 Step 1
           lstB.AddItem lstA.List(Index lstA)
           lstB.ItemData(lstB.NewIndex) = lstA.ItemData(Index lstA)
           If ChonThem = True Then
               TongGiaTienChoMotNguoi = TongGiaTienChoMotNguoi +
   lstA.ItemData(Index lstA)
           Else
               TongGiaTienChoMotNguoi = TongGiaTienChoMotNguoi -
   lstA.ItemData(Index lstA)
               lstGia.AddItem Str(lstA.ItemData(Index lstA))
           End If
       Next
       lstA.Clear
       If ChonThem = True Then
           lstGia.Clear
       End If
       lblTongGiaTien.Caption = TongGiaTienChoMotNguoi * SoThucKhach
   End Sub
   'Chức năng (6): Chọn tất cả.
   Private Sub cmdChonTatCa Click()
       ChuyenTatCa Tu ListBoxA Sang ListBoxB lstThucDon,
   lstMonDaChon, True
   End Sub
8. Nhấp đúp vào nút cmdBoChon và nhập code:
   'Chức năng (5): Bỏ chọn món.
   Private Sub cmdBoChon Click()
       Tu ListBoxA Sang ListBoxB lstMonDaChon, lstThucDon, False
   End Sub
9. Nhấp đúp vào nút cmdBoChonTatCa và nhập code:
   'Chức năng (7): Bỏ chọn tất cả.
   Private Sub cmdBoChonTatCa Click()
       ChuyenTatCa Tu ListBoxA Sang ListBoxB lstMonDaChon,
```

```
ChuyenTatCa_Tu_ListBoxA_Sang_ListBoxB lstMonDaChor
lstThucDon, False
End Sub
```

10. Nhấp đúp vào nút *cmdHoanTat* và nhập code:

```
'Chức năng (8): In thông báo các món đã chọn và tổng giá tiền.
Private Sub cmdHoanTat_Click()
Dim CacMonDaGoi As String
Dim i As Integer
CacMonDaGoi = ""
For i = 0 To lstMonDaChon.ListCount - 1 Step 1
CacMonDaGoi = CacMonDaGoi & Space(3) & i + 1 & ". " &
lstMonDaChon.List(i) & vbNewLine
Next
```

```
MsgBox "So luong thuc khach: " & SoThuckhach & VDNewLine &
vbNewLine & "Cac mon da duoc goi: " & vbNewLine & CacMonDaGoi &
vbNewLine & "Thanh tien: " & Format(TongGiaTienChoMotNguoi *
SoThucKhach, "# ### ###") & " dong" & vbNewLine & vbNewLine &
"Cam on quy khach!"
```

End

End Sub

11. Nhấp đúp vào nút *cmdChonLai* và nhập code:

```
'Chức năng (9): Trở lại trạng thái ban đầu để gọi món lại.
Private Sub cmdChonLai_Click()
lstThucDon.Clear
lstGia.Clear
lstMonDaChon.Clear
KhoiTao_lstThucDon
KhoiTao_lstGia
TongGiaTienChoMotNguoi = 0
lblTongGiaTien.Caption = "0"
End Sub
```

12. Lưu và nhấn nút Start hay F5 để chạy và xem kết quả:

Chương trình bắt đầu chạy:

💐. Dat mon an		
So thuc khach: 1	•	
Thuc don:	Gia (VND):	Cac mon da chon:
Nam kho tieu Nam chien gion Dau hu khia nuoc dua Dau hu chien xa Khoai tay xao nam	10000 15000 20000 15000 20000	>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>
cmdHoanTat	cmdChonLai	Thanh tien:

Nhấp đúp một món ăn trong danh sách Thực Đơn:

🔄. Dat mon an				
So thuc khach: 1	•			
Thuc don:	Gia (VND):		Cac mon da chon:	
Nam kho tieu Dau hu khia nuoc dua Dau hu chien xa Khoai tay xao nam Rau cu xao chua ngot	▲ 10000 20000 15000 20000 20000 ▼ 25000 ▼	>>>>	Nam chien gion	
cmdHoanTat	cmdChonLai		Thanh tien:	15000

Nhấp đúp thêm vài món nữa:

🔄. Dat mon an		
So thuc khach:	1 💌	
Thuc don:	Gia (VND):	Cac mon da chon:
Dau hu chien xa Rau cu xao chua ngot Rau muong xao Canh chua Canh rau cu	▲ 15000 25000 15000 20000 ▼ 25000 ▼	Nam chien gion Dau hu khia nuoc dua Khoai tay xao nam Canh nam
cmdHoanTat	cmdChonLai	Thanh tien:

Nhấp đúp vào danh sách Các món đã chọn để bỏ vài món:

🔄 Dat mon an		
So thuc khach: 1	•	
Thuc don:	Gia (VND):	Cac mon da chon:
Dau hu chien xa Rau cu xao chua ngot Rau muong xao Canh chua Canh rau cu	▲ 15000 25000 ■ 15000 ■ 20000 ■ 25000 ▼	Nam chien gion       Canh nam
cmdHoanTat	cmdChonLai	Thanh tien: 45000

Tăng số thực khách lên 4:

🔄, Dat mon an			
So thuc khach:	•		
Thuc don:	Gia (VND):		Cac mon da chon:
Dau hu chien xa Rau cu xao chua ngot Rau muong xao Canh chua Canh rau cu	▲ 15000 25000 15000 ■ 20000 ▼ 25000 ▼	>>>>	Nam chien gion Canh nam
cmdHoanTat	cmdChonLai		Thanh tien: 180000

Cuộn danh sách Thực đơn (danh sách Giá sẽ cuộn theo):

🕄, Dat mon an			3
So thuc khach: 4	<b>_</b>		
Thuc don:	Gia (VND):	Cac mon da chon:	
Rau muong xao Canh chua Canh rau cu Khoai tay xao nam Dau hu khia nuoc dua	▲ 15000 20000 ■ 25000 20000 ■ 20000 ▼ 20000 ▼	>> Nam chien gion >> Canh nam	
cmdHoanTat	cmdChonLai	Thanh tien: 180000	

Nhấp chọn vài món ăn trong Thực đơn (giữ phím Shift hoặc Ctrl và nhấn chuột):

🔄. Dat mon an		
So thuc khach:	4	
Thuc don:	Gia (VND):	Cac mon da chon:
Rau muong xao Canh chua	▲ 15000 20000 ▲	Nam chien gion     Canh nam
Canh rau cu Khoai tay xao nam	E 25000 20000	
Dau hu khia nuoc dua	20000 🔻	~
cmdHoanTat	cmdChonLai	Thanh tien: 180000

So thuc khach: 4	•		
Thuc don:	Gia (VND):		Cac mon da chon:
Nam kho tieu Dau hu chien xa Rau cu xao chua ngot Canh rau cu Khoai tay xao nam	10000 15000 25000 25000 20000	>>	Nam chien gion Canh nam Rau muong xao Canh chua Dau hu khia nuoc dua
cmdHoanTat	cmdChonLai		Thanh tien: 400000

Nhấn nút ">" để thêm món được chọn vào danh sách *Các món đã chọn*:

Chọn vài món trong danh sách Các món đã chọn và nhấn nút "<" để loại bỏ:

🔄, Dat mon an		
So thuc khach: 4	•	
Thuc don:	Gia (VND):	Cac mon da chon:
Nam kho tieu Dau hu chien xa Rau cu xao chua ngot Canh rau cu Khoai tay xao nam	10000 15000 25000 25000 20000	Nam chien gion     Canh nam     Rau muong xao     Canh chua     Canh chua     Canh ukhia nuoc dua
cmdHoanTat	cmdChonLai	Thanh tien: 400000

Dat mon an		
So thuc khach:	4 💌	
Thuc don:	Gia (VND):	Cac mon da chon:
Nam kho tieu Dau hu chien xa Rau cu xao chua ngot Canh rau cu Khoai tay xao nam	▲ 10000 15000 25000 ↓ 25000 ↓ 20000 ↓	>> Nam chien gion >> Rau muong xao Canh chua
cmdHoanTat	cmdChonLai	Thanh tien: 200000

Nhấn nút "<<" để bỏ chọn tất cả:

🔄, Dat mon an		
So thuc khach:	4 💌	
Thuc don:	Gia (VND):	Cac mon da chon:
Nam kho tieu Dau hu chien xa Rau cu xao chua ngot Canh rau cu Khoai tay xao nam	▲ 10000 15000 25000 ▼ 20000 ▼	
cmdHoanTat	cmdChonLai	Thanh tien:

Nhấn nút ">>" để chọn tất cả món:

🔄. Dat mon an			
So thuc khach:	4		
Thuc don:	Gia (VND):		Cac mon da chon:
		> 555 < <<	Nam kho tieu Dau hu chien xa Rau cu xao chua ngot Canh rau cu Khoai tay xao nam
cmdHoanTat	cmdChonLai		Thanh tien: 780000

Nhấn nút Chọn lại để thực hiện chọn lại:

🕄, Dat mon an		
So thuc khach:	4	
Thuc don:	Gia (VND):	Cac mon da chon:
Nam kho tieu Nam chien gion Dau hu khia nuoc dua Dau hu chien xa Khoai tay xao nam	10000     15000     20000     15000     20000     20000     •	>>> <<
cmdHoanTat	cmdChonLai	Thanh tien: 0

🔄. Dat mon an			
So thuc khach:	4		
Thuc don:	Gia (VND):		Cac mon da chon:
Dau hu chien xa Khoai tay xao nam Rau muong xao Canh chua Canh chua	▲ 15000 20000 15000 20000 25000 ▼	$\rightarrow$	Nam chien gion Dau hu khia nuoc dua Rau cu xao chua ngot Canh nam
cmdHoanTat	cmdChonLai		Thanh tien: 360000

Thực hiện chọn vài món và nhấn nút Hoàn tất:



# 7 Bài tập đề nghị

# 7.1 Bổ sung chức năng "Thay Thế" và "Thay Thế Tất Cả" cho bài 6.4

*Mô tả*: Bổ sung chức năng "Thay Thế", cho phép người dùng nhập vào một *Chuỗi thay thế* và thay chuỗi được tìm trong văn bản bằng chuỗi này.

Chức năng "Thay Thế Tất Cả": thực hiện thay thế tất cả các chỗi được tìm kiếm trong văn bản bằng *Chuỗi thay thế*.

# 7.2 Không xem hướng dẫn chi tiết và tự thực hiện lại bài 6.5

Mục đích: Giúp sinh viên có thể tự sử dụng thành thạo ListBox, ComboBox.

Hướng dẫn: Chỉ đọc phần Mô tả trong bài 6.5 và tự thực hiện viết code cho các chức năng.

\_\_\_\_\_HÊT\_\_\_\_\_